キャッシュON! リアルタイム高性能の世界をのぞく

なんて男らしい!

実験リサーチ! Linuxなし Cortex-A9プロセッサの実力

第4回　命令/データ・キャッシュを有効にする

中森 章

リスト1　命令キャッシュとデータ・キャッシュをONにする

```
    .text
    mov  r0,#0
    mcr  p15,0,r0,c7,c5,0       /* icache invalidate */
    mov  r0,#0x00
    mov  r5,#0x2000             /* dcache invalidate */
1:
    mov  r1,r0
    orr  r2,r0,#(1<<30)
    orr  r3,r0,#(2<<30)
    orr  r4,r0,#(3<<30)
    mcr  p15,0,r1,c7,c10,2
    mcr  p15,0,r2,c7,c10,2
    mcr  p15,0,r3,c7,c10,2
    mcr  p15,0,r4,c7,c10,2
    add  r0,r0,#0x20
    cmp  r0,r5
    blt  1b
    mrc  p15,0,r0,c1,c0,0               /* icache off */
    orr  r0,r0,#0x1000                  /* icache off */
    eor  r0,r0,#0x1000                  /* icache off */
    mcr  p15,0,r0,c1,c0,0               /* icache off */
    mrc  p15,0,r0,c1,c0,0               /* mmu off */
    orr  r0,r0,#0x1                     /* mmu off */
    eor  r0,r0,#0x1                     /* mmu off */
    mcr  p15,0,r0,c1,c0,0               /* mmu off */
    mrc  p15,0,r0,c1,c0,0               /* dcache off */
    orr  r0,r0,#0x4                     /* dcache off */
    eor  r0,r0,#0x4                     /* dcache off */
    mcr  p15,0,r0,c1,c0,0               /* dcache off */
    mov  r0,#0
    ldr  r0,=0xFFFFFFFF
    mcr  p15,0,r0,c3,c0,0       /* set access domain */
    ldr  r0,=0x20270000
    mcr  p15,0,r0,c2,c0,0                /* set TTBR */
    mov  r1,#0x00000000     /* start virtual address */
    mov  r2,#0x1000     /* the number of descriptors */
1:
    orr  r3,r1,#0xc00                      /* AP=11 */
    orr  r3,r3,#0x012                   /* C=0 B=0 */
    str  r3,[r0,r1, lsr #18]
                            /* set section descriptor */
    mcr  p15,0,r0,c7,c10,4      /* data sync barrier */
    add  r1,r1,#0x00100000
    subs r2,r2,#1
    bne  1b
    mcr  p15,0,r0,c8,c5,0          /* tlb invalidate */
    mrc  p15,0,r0,c1,c0,0        /* icache/dcache on */
    orr  r0,r0,#0x1000     /* icache/dcache on (I=1) */
    orr  r0,r0,#0x5      /* icache/dcache on (M=C=1) */
    mcr  p15,0,r0,c1,c0,0        /* icache/dcache on */
    bx   lr
```

ARMプロセッサの最大限の性能を引き出すには，キャッシュを動作させる必要があります．今回は，いよいよキャッシュを有効にします．

キャッシュを有効にしてみる

● 手順

前回（第3回，2015年3月号）までの知識があれば，命令キャッシュとデータ・キャッシュを使うことができるはずです．その手順を以下に示します．

(1) MMUがOFF，データ・キャッシュがOFFの状態で，MMU関連の設定を行う．
(2) システム制御レジスタのMビット，Cビット，Iビットを1に設定する．

● アセンブラで手順を書く

これらの手順を示すアセンブラでのプログラムを**リスト1**に示します．アドレス変換にセクションを使用し，仮想アドレスと物理アドレスが等しくなるように設定しています．0x00000000番地から0xFFF00000番地で始まるセクションの設定をしますが，セクション・テーブルをプログラムで自動生成しています．これらのテーブルは，あらかじめメモリ上に格納しておきます．そのテーブルの先頭アドレス（物理アドレス）をTTBRに設定するのが普通ですが，テーブルがROMなどのライト不可能な領域にない場合（書き込み可能な場合）はプログラムで生成した方が簡単に思えます．

ここで，セクション・テーブルのセクション記述子ではC=0，B=0，つまり，全領域を非キャッシュに指定しています．こうしないと，U-Boot内で矛盾が起こる箇所があるらしく，ハングアップしてしまいます．

ところで，**リスト1**にはセクション・テーブルをメモリにライトした後，データ同期化バリアを実行しています．これは，ライト・バッファなど，CPU内に